
1. [bookmark: _Toc122434485][bookmark: _Toc107322943]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 8.1.6.2.3 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
[bookmark: _Hlk90298041]Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

0. [bookmark: _Toc107322944]Verification Test Summary 
Test Case: 	8.1.6.2.3
ATS Version:	iWD-TTCN3-B2020-09_D22wk23
System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR
UE used:	Mediatek MT6983
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc107322945]Corrections required
1. [bookmark: _Toc54888063][bookmark: _Toc93508300][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc107322946]Change 1
Can be removed if MCC160 approves R5s220801 change 1.
	Function name
	templatecs_CoOrdEUTRASysInfo()

	Reason for change
	This test is required to pass information of PhysCellId to NR component as NR has to evaluate UE reported MeasResultNeighCellListEUTRA on UEInformationResponse at step15. So cs_CoOrdEUTRASysInfo template and associated its type record should be modified.

	Summary of change
	Check below for the changes

	TTCN module
	CommonIratDefs.ttcn

	MCC160 Comment
	



Before Change:
	  type record CoOrd_EUTRASysInfo_Type {         /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    integer Arfcn,
    integer DL_Bandwidth,                        // @sic R5s100871 Need to pass this through the co-ord msg sic@
    EUTRA_HOInfo_Type  HO optional,                     // extended for HO test cases
    integer CA_Tested_Integer optional,
    O3_Type  PLMN optional,                      // the PLMN of the cell, defined in octets
    B28_Type CellId optional,                    // the EUTRA cell identity
    B16_Type TAC  optional,
    integer  freqBandIndicator  optional,
    boolean  UL_CA              optional,        // @sic R5-183277 sic@
    integer  UL_Arfcn           optional         // @sic R5-217955 sic@
  };
       
 template (value) CoOrd_EUTRASysInfo_Type cs_CoOrdEUTRASysInfo(integer p_Arfcn,
                                                                integer p_DLBW,
                                                                template (omit) EUTRA_HOInfo_Type p_HOInfo := omit,
                                                                template (omit) integer p_CATested := omit,
                                                                template (omit) O3_Type  p_PLMN := omit,
                                                                template (omit) B28_Type p_CellId := omit,
                                                                template (omit) B16_Type p_TAC  := omit,
                                                                template (omit) integer p_FreqBandIndicator := omit,
                                                                template (omit) boolean p_UL_CA := omit,
                                                                template (omit) integer p_UL_Arfcn := omit) :=
  { /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC_IRAT, POS) */
    Arfcn :=  p_Arfcn,
    DL_Bandwidth := p_DLBW,
    HO    := p_HOInfo,
    CA_Tested_Integer := p_CATested,
    PLMN := p_PLMN,
    CellId := p_CellId,
    TAC := p_TAC,
    freqBandIndicator := p_FreqBandIndicator,
    UL_CA := p_UL_CA,
    UL_Arfcn :=  p_UL_Arfcn    // @sic R5-217955 sic@
//added PhysCellId here
  };





After Change:
	     type record CoOrd_EUTRASysInfo_Type {         /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    integer Arfcn,
    integer DL_Bandwidth,                        // @sic R5s100871 Need to pass this through the co-ord msg sic@
    EUTRA_HOInfo_Type  HO optional,                     // extended for HO test cases
    integer CA_Tested_Integer optional,
    O3_Type  PLMN optional,                      // the PLMN of the cell, defined in octets
    B28_Type CellId optional,                    // the EUTRA cell identity
    B16_Type TAC  optional,
    integer  freqBandIndicator  optional,
    boolean  UL_CA              optional,        // @sic R5-183277 sic@
    integer  UL_Arfcn           optional,         // @sic R5-217955 sic@
    integer PhysCellId          optional
  };  

 template (value) CoOrd_EUTRASysInfo_Type cs_CoOrdEUTRASysInfo(integer p_Arfcn,
                                                                integer p_DLBW,
                                                                template (omit) EUTRA_HOInfo_Type p_HOInfo := omit,
                                                                template (omit) integer p_CATested := omit,
                                                                template (omit) O3_Type  p_PLMN := omit,
                                                                template (omit) B28_Type p_CellId := omit,
                                                                template (omit) B16_Type p_TAC  := omit,
                                                                template (omit) integer p_FreqBandIndicator := omit,
                                                                template (omit) boolean p_UL_CA := omit,
                                                                template (omit) integer p_UL_Arfcn := omit,
                                                                template (omit) integer p_PhysCellId := omit) :=
  { /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC_IRAT, POS) */
    Arfcn :=  p_Arfcn,
    DL_Bandwidth := p_DLBW,
    HO    := p_HOInfo,
    CA_Tested_Integer := p_CATested,
    PLMN := p_PLMN,
    CellId := p_CellId,
    TAC := p_TAC,
    freqBandIndicator := p_FreqBandIndicator,
    UL_CA := p_UL_CA,
    UL_Arfcn :=  p_UL_Arfcn,    // @sic R5-217955 sic@
    PhysCellId := p_PhysCellId
  };


1. [bookmark: _Toc107322947]Change 2
	Function name
	Funtion f_TC_8_1_6_2_3_NR5GC_EUTRA()

	Reason for change
	1. Preamble on EUTRA not required, as prose expecting preamble on NR.
2. Need to power up EUTRA cell once UE is done with preamble on EUTRA cell, so wait for NR component is required.

	Summary of change
	Check below for the changes

	TTCN module
	RRC_MDT_InterRAT_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change:
	          function f_TC_8_1_6_2_3_NR5GC_EUTRA() runs on EUTRA_Multi5GS_PTC
  { // Inter-RAT MDT / Radio Link Failure / Reporting at E-UTRA Inter-RAT handover
    
    f_EUTRA38_IRAT_Init(c31); //System information combination 31 as defined in TS 36.508 [7] clause 4.4.3.1 is used in the E-UTRA cell.
    //initiate the power level
    f_EUTRA_CellInfo_InitMaxReferencePower( eutra_Cell1, -73);
    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    
    f_EUTRA38_NR5GC_PreambleAndSwitchOff (eutra_Cell1);
    f_EUTRA38_SetCellPower(eutra_Cell1, -73, tsc_NonSuitableOffCellRS_EPRE);// @sic R5-216499 R5-220052 sic@
    
    f_EUTRA_TestBody_Set(true);
    fl_TC_8_1_6_2_3_NR5GC_EUTRA_TestBody ();
    f_EUTRA_TestBody_Set(false);

    // UE now on NR
    f_EUTRA_ReleaseAllCells();
  }



After Change:
	  import from EUTRA_CellInfoInit all;    
  function f_TC_8_1_6_2_3_NR5GC_EUTRA() runs on EUTRA_Multi5GS_PTC
  { // Inter-RAT MDT / Radio Link Failure / Reporting at E-UTRA Inter-RAT handover
    
    f_EUTRA38_IRAT_Init(c31); //System information combination 31 as defined in TS 36.508 [7] clause 4.4.3.1 is used in the E-UTRA cell.
    //initiate the power level
    f_EUTRA_CellInfo_InitMaxReferencePower( eutra_Cell1, -73);
    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    //f_EUTRA38_NR5GC_PreambleAndSwitchOff (eutra_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);
    f_EUTRA38_SetCellPower(eutra_Cell1, -73, tsc_NonSuitableOffCellRS_EPRE);// @sic R5-216499 R5-220052 sic@
    
    f_EUTRA_TestBody_Set(true);
    fl_TC_8_1_6_2_3_NR5GC_EUTRA_TestBody ();
    f_EUTRA_TestBody_Set(false);

    // UE now on NR
    f_EUTRA_ReleaseAllCells();
  }


1. [bookmark: _Toc107322948]Change 3
	Function name
	function fl_TC_8_1_6_2_4_NR5GC_EUTRA_TestBody()

	Reason for change
	1. System Co-ordination required to send EUTRA cell information like physcellID and frequency information to NR test body. 
2. TAU process along with security activation required also need to Establishment of default and dedicated bearers.

	Summary of change
	Check below for the changes

	TTCN module
	RRC_MDT_InterRAT_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_8_1_6_2_3_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var octetstring v_NR_HOCmd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    // Send EUTRA frequency to NR
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    // @siclog "Step 5" siclog@
    f_EUTRA38_TrackingAreaUpdateFromN1 (eutra_Cell1,tsc_RRC_TI_Def,-,-,noRrcConnectionRelease);// @sic R5-223273 sic@
    f_EUTRA38_SetCellPower(eutra_Cell1, -85, tsc_NonSuitableOffCellRS_EPRE);// @sic R5-216499 R5-220052 sic@
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Power Level"));
    // @siclog "Step 7" siclog@
    // the source MME shall increase its locally kept NCC value by one and compute a fresh NH using A.4
    v_SecurityParams.NH := f_EUTRA_NB_Authentication_S12(v_SecurityParams.KDF,
                                                         v_SecurityParams.KASME,
                                                         v_SecurityParams.NH);
    // 33.501 cl. 8.4.2 point 3: the target AMF shall also set the NCC to zero
    v_SecurityParams.NCC := 2; // @sic R5-212374 sic@
    f_EUTRA_Security_Set (v_SecurityParams);
    // Send latest parameters to NR
    f_EUTRA38_Send_AuthParametersToNR();
    
    f_EUTRA38_SendPduSessionInfoListToNR();
    
    // Wait for NR to send back encoded RRC Reconfiguration message
    v_NR_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(NR, "HO Cmd");

    SRB.send( cas_SRB1_RrcPdu_REQ( eutra_Cell1, cs_TimingInfo_Now, cs_MobilityFromEutraCommand_Handover( tsc_RRC_TI_Def,
                                                                                                         false,
                                                                                                         nr,
                                                                                                         v_NR_HOCmd,
                                                                                                         omit)));
    // UE now moved to NR
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");
  }



After Change:
	      function fl_TC_8_1_6_2_3_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var octetstring v_NR_HOCmd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;
    var NasCount_Type v_NasCountUL;
    
    // Send EUTRA frequency to NR
    //f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    v_CoOrd_SysInfo := cs_EUTRA_NR_CoOrdSysInfo_Eutra({f_EUTRA_CoOrdSysInfo(eutra_Cell1)}); //this one returns only dlarfcn and bw
    v_CoOrd_SysInfo.Eutra[0].PhysCellId := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);   //adds the phycellID to the coorData
    f_EUTRA_NR_SendSysinfoCoOrd(NR, v_CoOrd_SysInfo);
    
    // @siclog "Step 5" siclog@

    //v_NasCountUL := f_EUTRA38_TrackingAreaUpdateFromN1 (eutra_Cell1,tsc_RRC_TI_Def, omit, -, noRrcConnectionRelease);
    fl_TAU_N1ToS1(eutra_Cell1, tsc_RRC_TI_Def, "Step5", "Step6", omit);
    v_SecurityParams := f_EUTRA_Security_Get();

    

    f_EUTRA38_SetCellPower(eutra_Cell1, -85, tsc_NonSuitableOffCellRS_EPRE);// @sic R5-216499 R5-220052 sic@
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Power Level"));
    // @siclog "Step 7" siclog@
    // the source MME shall increase its locally kept NCC value by one and compute a fresh NH using A.4
    v_SecurityParams.NH := f_EUTRA_NB_Authentication_S12(v_SecurityParams.KDF,
                                                         v_SecurityParams.KASME,
                                                         v_SecurityParams.NH);
    // 33.501 cl. 8.4.2 point 3: the target AMF shall also set the NCC to zero
    v_SecurityParams.NCC := 2; // @sic R5-212374 sic@
    f_EUTRA_Security_Set (v_SecurityParams);
    // Send latest parameters to NR
    f_EUTRA38_Send_AuthParametersToNR();
    
    f_EUTRA38_SendPduSessionInfoListToNR();
    
    // Wait for NR to send back encoded RRC Reconfiguration message
    v_NR_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(NR, "HO Cmd");

    SRB.send( cas_SRB1_RrcPdu_REQ( eutra_Cell1, cs_TimingInfo_Now, cs_MobilityFromEutraCommand_Handover( tsc_RRC_TI_Def,
                                                                                                         false,
                                                                                                         nr,
                                                                                                         v_NR_HOCmd,
                                                                                                         omit)));
    // UE now moved to NR
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");
  }


1. [bookmark: _Toc107322949]Change 4
	Function name
	function fl_TAU_N1ToS1 added

	Reason for change
	1. TAU process along with security activation added as prose expecting UE Tracking area updating / inter-system change from N1 mode to S1 mode in EMM-IDLE mode

	Summary of change
	Check below for the changes

	TTCN module
	RRC_MDT_InterRAT_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change:
	    function fl_TAU_N1ToS1 (EUTRA_CellId_Type p_CellId,
                          RRC_TransactionIdentifier p_RRC_TI,
                          charstring p_StepNo1 := "",
                          charstring p_StepNo2 := "",
                          template EUTRA_CellId_Type p_CellIdPreviouslyCampedOn := eutra_Cell_NonSpecific,
                          boolean p_FirstN1_S1Transition := true) runs on EUTRA_Multi5GS_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var EUTRA38_DRBInfoList_Type v_ExistingDRBs;
    var template (value) RadioBearerList_Type v_DrbConfigList;
    var DRB_ToAddModList v_DRB_ToAddModList;
    var template (omit) SRB_ToAddModList v_SRB_ToAddModList := {cs_508_SRB2_ToAddMod_DEFAULT};
    var integer i;
    
    v_NasInd := f_EUTRA38_TAUReqFromN1 (p_CellId, p_RRC_TI, -, -, p_CellIdPreviouslyCampedOn, -, p_FirstN1_S1Transition); // @sic R5s201062 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_StepNo1);
    
    // @sic R5-204399, R5s201465 sic@ get all DRBs transferred from NR
    v_ExistingDRBs := f_EUTRA38_MobileInfo_GetDRBInfoList();
    for (i:=0;  i < lengthof(v_ExistingDRBs); i :=i+1) {
      v_DrbConfigList[i] := v_ExistingDRBs[i].SS_Config;
    }
                                           
    v_SecurityParams := f_EUTRA_Security_Get();
    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_SecurityParams, v_NasInd.SecurityProtection.NasCount, v_DrbConfigList); // @sic R5s200878, R5-204399 sic@
    f_EUTRA_Security_Set(v_SecurityParams); //Saving security parameters
    
    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(p_CellId, v_NasInd, -, noRrcConnectionRelease, -, -, -, -,true); // @sic R5s220006 sic@

    // @sic R5s201465 sic@ get all DRBs again in case more have been established
    v_ExistingDRBs := f_EUTRA38_MobileInfo_GetDRBInfoList();
    for (i:=0;  i < lengthof(v_ExistingDRBs); i :=i+1) {
      v_DRB_ToAddModList[i] := v_ExistingDRBs[i].DRB_ToAdd;
    }
                                           
    f_EUTRA_RRCConnectionReconfiguration_Send(p_CellId,         // @sic R5s211690 sic@
                                              p_RRC_TI,
                                              0,
                                              v_SRB_ToAddModList,
                                              v_DRB_ToAddModList);

    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId,
                                                  p_RRC_TI);

      }



1. [bookmark: _Toc107322950]Change 5
	Function name
	function fl_TC_8_1_6_2_3_TestBody()

	Reason for change
	1. Activated Security on EUTRA cell can be derived after establishing TAU on EUTRA cell so moved to step 7 .
2. DRB release and configuration is required once UE moved to EUTRA cell.
3. No need to bit to integer conversion on v_CoOrd_SysInfo.Eutra[0].CellId as it will return EUTRA_PhysCellId type.
4. PDu session modification not required
5. Parameters to be passed on got interchnaged on MeasResultNeighCellListEUTRA at step 14.
6. As per proposed prose CR “set the nrFailedPCellId in failedPCellId to the global cell identity and the tracking area code, if available, and otherwise to the physical cell identity and carrier frequency of the PCell where radio link failure is detected;” and “set timeUntilReconnection in VarRLF-Report of TS 38.331 [82] to the time that elapsed since the last radio link failure or handover failure;” and “set eutraReconnectCellId in reconnectCellId in VarRLF-Report of TS 38.331 [82] to the global cell identity and the tracking area code of the PCell;” and “timeConnFailure
This field is used to indicate the time elapsed since the last HO initialization until connection failure. Actual value = field value * 100ms. The maximum value 1023 means 102.3s or longer.” And “set the noSuitableCellFound in the VarRLF-Report to true;”
7. As procedure 4.5A.2C.2.2-1 is for “UE-requested PDU session modification after the first S1 to N1 mode change”, but in the preamble UE is already in state 3N-A. Therefore step 12 could be avoid.

	Summary of change
	Check below for the changes

	TTCN module
	RRC_MDT_IntraNR_RLF_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_8_1_6_2_3_TestBody() runs on NR5GC_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, -90, tsc_NR_NonSuitableOffCellSSS_EPRE)};//@sic R5-216499 R5-220052 sic@
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1,tsc_NR_NonSuitableOffCellSSS_EPRE)};//@sic R5-216499 R5-220052 sic@
    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;
    var EUTRA_PhysCellId v_EUTRA_CellId;
    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;
    var Hysteresis v_Hysteresis := 0; //@sic R5-218335 sic@
    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;
    var RSRP_Range v_B2_Threshold1 := 73;
    var RSRP_RangeEUTRA v_B2_Threshold2EUTRA := 58;
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;
    var RLF_Report_r16 v_Nr_Rlf_Report_r16;
    var default v_MyDefaultVar := null;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
      
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;
    var template (omit) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var template (present) NR_RLF_Report_measResultNeighCells v_MeasResultNeighCells_r16;
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);
    var PhysCellId v_PhysCellId_Cell1 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    var ARFCN_ValueNR v_SSB_Freq_Cell1 := f_NR_CellInfo_GetFrequencySSB(nr_Cell1);
    var template (present) UL_DCCH_Message v_RRCConnReconfigComplete;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var integer i;
         
    //Power levels of NR cell 1: Table 8.1.6.2.3.3.2-1, 8.1.6.2.3.3.2-2
    /*         |    FR1     |    FR2
     * -------------------------------
     *   T0    |    -90     |    FFS----//@sic R5-216499 R5-220052 sic@
     * -------------------------------
     *   T1    |    off     |    off
     * -------------------------------
     *   T2    |    -88     |    FFS   //@sic R5-216499 R5-220052 sic@.
     * -------------------------------
     */
    v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo (EUTRA);
    
    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;
    v_EUTRA_CellId := bit2int (v_CoOrd_SysInfo.Eutra[0].CellId);
    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth);

    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1-2" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B2
    //The UE transmits an RRCReconfigrationComplete message
    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};
    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};
    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB2_EUTRA(v_B2_Threshold1, v_B2_Threshold2EUTRA, v_Hysteresis, v_ReportAmount, true, false))};
    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};
    v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_38508_MeasGapConfig_UE(cs_38508_GapConfig), cs_38508_QuantityConfig_EUTRA);
    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1,v_MeasConfig);

    //@siclog "Step 3" siclog@
    //The UE transmits a MeasurementReport message on NR Cell 1.
    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId_Cell1, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp_Rsrq)}));

    //@siclog "Step 4" siclog@
    //The SS changes NR Cell1 and E-UTRA Cell 1 parameters according to the row "T1" in table 8.1.6.2.3.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 5" siclog@
    //Generic test procedure in TS 38.508-1 Table 4.9.7.2.2-1 is performed on E-UTRA Cell 1.
    f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);
    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Power Level");
// added function for DRB release and config
   
 //@siclog "Step 6" siclog@
    //The SS changes NR Cell 1 parameter according to the row "T2" in Table 8.1.6.2.3.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    //The SS transmits a MobilityFromEUTRACommand message on E-UTRA Cell 1.
//get security context from EUTRAN
    f_NR5GC_MappedContextFromS1_InitNAS (v_SecurityRx, true);

    // this also sends octetstring to EUTRA and waits for the Complete message
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit a RRCReconfigurationComplete message containing rlf-InfoAvailable on NR Cell 1?
    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);
    v_RRCConnReconfigComplete:=cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-,cr_RRCReconfigurationComplete_v1560(-,cr_RRCReconfigurationComplete_v1610_RLF)));
    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList, v_SecurityRx,-,-,v_RRCConnReconfigComplete);

    //@siclog "Step 9" siclog@
    //The UE transmits a ULInformationTransfer message and a REGISTRATION REQUEST message indicating "mobility registration updating"? is sent to update the registration of the actual tracking area.
    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected))) -> value v_ReceivedAsp;
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }

    //@siclog "Step 10-11" siclog@
    //SS sends an DLInformationTransfer message and a REGISTRATION ACCEPT message containing a 5G-GUTI.
    //The UE transmits an ULInformationTransfer message and a REGISTRATION COMPLETE message.
    f_NR5GC_MobileInfo_SetGMM_MobilityInfo(v_GMM_MobilityInfo);
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, -, tsc_NR_RbId_SRB2);
    // Clear store of msgs received in parallel
    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq(); // @sic R5s220174 sic@
    // Then activate store for any new msgs received - this works for Modification Reqs too
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell1));
    // make sure all the PDNs configured on EUTRA are changed in routing table
    for (i := 0; i < lengthof (v_PduSessionInfoList); i := i + 1) {
      f_IP_ChangeDrbMappingNR(IP, nr_Cell1, cs_IP_QosFlowId(v_PduSessionInfoList[i].PDU_SessionId, v_PduSessionInfoList[i].QFI), v_PduSessionInfoList[i].PdnIndex);
    }

    //@siclog "Step 12" siclog@
    //The procedure in table 4.5A.2C.2.2-1 in TS 38.508-1 [4] is performed.
    f_NR5GC_IRAT_S1ToN1_SessionModification(nr_Cell1, v_PduSessionInfoList, -, tsc_NR_RbId_SRB2);
    deactivate(v_MyDefaultVar);

    //@siclog "Step 13" siclog@
    //The SS transmits a UEInformationRequest message.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                    -,
                                    cs_38508_UEInformationRequest(tsc_NR_RRC_TI_Def, -, -, -, -, true_)));

    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit a UEInformationResponse message on NR Cell 1?
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1,
                                       cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def,-,-,-,-,-,?))) -> value v_ReceivedAsp;
    v_Nr_Rlf_Report_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16;
    v_MeasResultNeighCells_r16:=cr_NR_RLF_Report_measResultNeighCells_Eutra_r16(cr_MeasResultNeighCellListEUTRA_1Entry(v_EUTRA_CellId,v_ARFCN_ValueEUTRA,
                                                                                                                        f_EUTRA_Get_MeasQuantityResults()));
    if (match(v_Nr_Rlf_Report_r16, cr_NR_Rlf_Report(f_NR_CellInfo_GetRNTI(nr_Cell1), -, //SSB omit incorrect
                                                    ?, ?,
                                                    v_MeasResultNeighCells_r16, omit,
                                                    cr_NR_RLF_Report_failedPCellId_PciArfcn_r16(v_PhysCellId_Cell1, v_SSB_Freq_Cell1),
                                                    omit, omit, omit,
                                                    ?, ?,
                                                    rlf,
                                                    t310_Expiry,
                                                    -, omit)))// noSuitableCellFound-r16-omit?
                                                    {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
    }
  }



After Change:
	    function fl_TC_8_1_6_2_3_TestBody() runs on NR5GC_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, -90, tsc_NR_NonSuitableOffCellSSS_EPRE)};//@sic R5-216499 R5-220052 sic@
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1,tsc_NR_NonSuitableOffCellSSS_EPRE)};//@sic R5-216499 R5-220052 sic@
    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;
    var EUTRA_PhysCellId v_EUTRA_CellId;
    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;
    var Hysteresis v_Hysteresis := 0; //@sic R5-218335 sic@
    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;
    var RSRP_Range v_B2_Threshold1 := 73;
    var RSRP_RangeEUTRA v_B2_Threshold2EUTRA := 58;
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;
    var RLF_Report_r16 v_Nr_Rlf_Report_r16;
    var default v_MyDefaultVar := null;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;
    var template (omit) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var template (present) NR_RLF_Report_measResultNeighCells v_MeasResultNeighCells_r16;
    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;// := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);
    var PhysCellId v_PhysCellId_Cell1 := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    var ARFCN_ValueNR v_SSB_Freq_Cell1 := f_NR_CellInfo_GetFrequencySSB(nr_Cell1);
    var template (present) UL_DCCH_Message v_RRCConnReconfigComplete;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var integer i;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(nr_Cell1);
    var RNTI_Value_Type v_C_RNTI_beforeRLF;
    var PLMN_Identity v_PLMN_Identity_Cell1 := f_NR_CellInfo_GetPLMN (nr_Cell1);
    var CellIdentity v_CellIdentity_Cell1 := f_NR_CellInfo_GetCellIdentity(nr_Cell1);
    var TrackingAreaCode v_TAC_Cell1 := f_NR_CellInfo_GetTAC (nr_Cell1);

    //Power levels of NR cell 1: Table 8.1.6.2.3.3.2-1, 8.1.6.2.3.3.2-2
    /*         |    FR1     |    FR2
     * -------------------------------
     *   T0    |    -90     |    FFS----//@sic R5-216499 R5-220052 sic@
     * -------------------------------
     *   T1    |    off     |    off
     * -------------------------------
     *   T2    |    -88     |    FFS   //@sic R5-216499 R5-220052 sic@.
     * -------------------------------
     */
    v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo (EUTRA);
    
    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;
    v_EUTRA_CellId := v_CoOrd_SysInfo.Eutra[0].PhysCellId;
    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth);
    v_C_RNTI_beforeRLF := f_NR_CellInfo_GetRNTI(nr_Cell1);

    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1-2" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B2
    //The UE transmits an RRCReconfigrationComplete message
    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};
    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};
    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB2_EUTRA(v_B2_Threshold1, v_B2_Threshold2EUTRA, v_Hysteresis, v_ReportAmount, true, false))};
    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};
    v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_38508_MeasGapConfig_UE(cs_38508_GapConfig), cs_38508_QuantityConfig_EUTRA);
    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1,v_MeasConfig);

    //@siclog "Step 3" siclog@
    //The UE transmits a MeasurementReport message on NR Cell 1.
    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId_Cell1, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp_Rsrq)}));

    //@siclog "Step 4" siclog@
    //The SS changes NR Cell1 and E-UTRA Cell 1 parameters according to the row "T1" in table 8.1.6.2.3.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 5" siclog@
    //Generic test procedure in TS 38.508-1 Table 4.9.7.2.2-1 is performed on E-UTRA Cell 1.
    f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);
    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Power Level");
    f_NR_SS_SRBs_DRBs_Release(nr_Cell1, -, v_DrbIdList); //let the SS know that there was OOS
    f_NR_SS_SRBs_DRBs_Config(nr_Cell1, -, omit); //recpn

    //@siclog "Step 6" siclog@
    //The SS changes NR Cell 1 parameter according to the row "T2" in Table 8.1.6.2.3.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    //The SS transmits a MobilityFromEUTRACommand message on E-UTRA Cell 1.
    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);
    f_NR5GC_MappedContextFromS1_InitNAS (v_SecurityRx, true);

    // this also sends octetstring to EUTRA and waits for the Complete message
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit a RRCReconfigurationComplete message containing rlf-InfoAvailable on NR Cell 1?
    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);
    v_RRCConnReconfigComplete:=cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-,cr_RRCReconfigurationComplete_v1560(-,cr_RRCReconfigurationComplete_v1610_RLF)));
    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList, v_SecurityRx,-,-,v_RRCConnReconfigComplete);

    //@siclog "Step 9" siclog@
    //The UE transmits a ULInformationTransfer message and a REGISTRATION REQUEST message indicating "mobility registration updating"? is sent to update the registration of the actual tracking area.
    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected))) -> value v_ReceivedAsp;
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }

    //@siclog "Step 10-11" siclog@
    //SS sends an DLInformationTransfer message and a REGISTRATION ACCEPT message containing a 5G-GUTI.
    //The UE transmits an ULInformationTransfer message and a REGISTRATION COMPLETE message.
    f_NR5GC_MobileInfo_SetGMM_MobilityInfo(v_GMM_MobilityInfo);
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, -, tsc_NR_RbId_SRB2);
    // Clear store of msgs received in parallel
    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq(); // @sic R5s220174 sic@
    // Then activate store for any new msgs received - this works for Modification Reqs too
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell1));
    // make sure all the PDNs configured on EUTRA are changed in routing table
    for (i := 0; i < lengthof (v_PduSessionInfoList); i := i + 1) {
      f_IP_ChangeDrbMappingNR(IP, nr_Cell1, cs_IP_QosFlowId(v_PduSessionInfoList[i].PDU_SessionId, v_PduSessionInfoList[i].QFI), v_PduSessionInfoList[i].PdnIndex);
    }

    //@siclog "Step 12" siclog@
    //The procedure in table 4.5A.2C.2.2-1 in TS 38.508-1 [4] is performed.
    //f_NR5GC_IRAT_S1ToN1_SessionModification(nr_Cell1, v_PduSessionInfoList, -, tsc_NR_RbId_SRB2);
    deactivate(v_MyDefaultVar);

    //@siclog "Step 13" siclog@
    //The SS transmits a UEInformationRequest message.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                    -,
                                    cs_38508_UEInformationRequest(tsc_NR_RRC_TI_Def, -, -, -, -, true_)));

    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit a UEInformationResponse message on NR Cell 1?
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1,
                                       cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def,-,-,-,-,-,?))) -> value v_ReceivedAsp;
    v_Nr_Rlf_Report_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16;
    v_MeasResultNeighCells_r16:=cr_NR_RLF_Report_measResultNeighCells_Eutra_r16(cr_MeasResultNeighCellListEUTRA_1Entry(v_ARFCN_ValueEUTRA,v_EUTRA_CellId,
                                                                                                                        f_EUTRA_Get_MeasQuantityResults()));                                                                                                                   
    
    if (match(v_Nr_Rlf_Report_r16, cr_NR_Rlf_Report(v_C_RNTI_beforeRLF, -, //SSB omit incorrect
                                                    ?, ?,
                                                    v_MeasResultNeighCells_r16, omit,
                                                    (cr_NR_RLF_Report_failedPCellId_cellGlobalId_r16(cr_CGI_Info_Logging_r16(v_PLMN_Identity_Cell1, v_CellIdentity_Cell1, v_TAC_Cell1))),
                                                    cr_eutraReconnectCellId_r16, ?, omit,
                                                    -, ?,
                                                    rlf,
                                                    t310_Expiry,
                                                    -, omit)))// noSuitableCellFound-r16-omit?
                                                    {
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
    }
  }




1. [bookmark: _Toc107322951]Change 6
	Function name
	Added template cr_eutraReconnectCellId_r16

	Reason for change
	1. Added Template to evaluate eutraReconnectCellId_r16 reprted by UE on reconnectCellId in UEInformationResponse. 

	Summary of change
	Check below for the changes

	TTCN module
	NR5GC_MDT_Specific_Templates.ttcn

	MCC160 Comment
	



Before Change:
	            template NR_RLF_Report_reconnectCellId cr_eutraReconnectCellId_r16 :=
  { /* @status APPROVED (NR5GC) */
      eutraReconnectCellId_r16 := ?
  };



1. [bookmark: _Toc107322952]Branches executed 
[bookmark: _Hlk90298239][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]On Anritsu, the test case was executed on NR band n78 using NR ciphering nea2 and integrity algorithm nia2.

3 [bookmark: _Toc107322953]Execution Log Files 
3.1 [bookmark: _Toc93508305][bookmark: _Toc107322954] Mediatek MT6983
The Mediatek MT6983UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Anritsu\ TC_8_1_6_2_3_LOG.xml
· PICS/PIXIT parameter file:
Anritsu\ TC_8_1_6_2_3_PIXIT.txt


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc107322955]References 
	[1]
	R5s220832 - Supporting information for addition of NR5GC Inter-RAT MDT test case 8.1.6.2.3 in FR1




